```
/*******************************************************************
*
*Incorporation of time-varying coefficients into models 2, 3 and 4
*
*******************************************************************/
*Set up log and working directory
capture log close
version 13.1
set linesize 100
set more off

cd "C:\data\malaria\results"
loc today = c(current_date)
log using "malariaproject_log_`today'.txt", append text

*cox model ties handling
loc ties efron

* prepare folder for results
local T = c(current_time)
local T = subinstr("`T'",":","_",.)
mkdir "`ties' `today' `T'"
cd "`ties' `today' `T'"

****note '3' and '7' refer to days following presentation (ie. _t = 4 and _t = 8, respectively)

*Model 2
/*load data*/
use "C:\data\malaria\MalEps_v1.9.3_r5oct2015.dta", clear
/*create indicator variables*/
xi i.SpeciesX i.EthnicX i.AGR4_4b i.sexPreg i.oral_v_dhp
stset AdmFU15, fail(AdmNext14) id(obsno)
/*split time*/
stsplit new, at(4 8)  /* Splitting time (see note aboe about t4 and t8 above) */


loc varlist _ISpeciesX_2    /// /*List of variables to split */
            _ISpeciesX_4    ///
            _ISpeciesX_5    ///
            _IEthnicX_2     ///
            _IEthnicX_3     ///
            _IAGR4_4b_1     ///
            _IAGR4_4b_2     ///
            _IAGR4_4b_3     ///
            _IsexPreg_2     ///
            _IsexPreg_3     ///
            _Ioral_v_dh_1

/*generate interaction term*/
foreach i of varlist `varlist' {
gen tv3`i' = `i' * (new==4)
gen tv7`i' = `i' * (new==8)
}

/*List of variables for model including interactions with time */
loc varlist _ISpeciesX_2    tv3_ISpeciesX_2     tv7_ISpeciesX_2     ///
            _ISpeciesX_4                                            ///
            _ISpeciesX_5                                            ///
            _IEthnicX_2                                             ///
            _IEthnicX_3    tv3_IEthnicX_3       tv7_IEthnicX_3      ///
            _IAGR4_4b_1                                             ///
            _IAGR4_4b_2     tv3_IAGR4_4b_2       tv7_IAGR4_4b_2     ///
            _IAGR4_4b_3                         tv7_IAGR4_4b_3      ///
            _IsexPreg_2                         tv7_IsexPreg_2      ///
            _IsexPreg_3     tv3_IsexPreg_3      tv7_IsexPreg_3      ///
            _Ioral_v_dh_1


stcox `varlist' if ip==0, efron allbaselevels vsquish cluster(hrn) cformat(%6.2f)

/*save and store estimates and scaled Schoenfeld residuals for later access*/
estimates
estimates store M2aTV37_24oct2015
estimates save M2aTV37_24oct2015
predict sch_M2aTV37*, sca
save MalEps_v1.9.3_M2a_TV37.dta
linktest, cluster(hrn) efron
estat phtest, d

lincom _b[_ISpeciesX_2]+_b[tv3_ISpeciesX_2]+ _b[tv7_ISpeciesX_2], eform
lincom _b[_ISpeciesX_2]+_b[tv3_ISpeciesX_2], eform

*Model 3
set more off
*load data
use "C:\data\malaria\MalEps_v1.9.3_r5oct2015.dta", clear
```

```
/*create indicator variables*/
xi i.SpeciesX i.EthnicX i.AGR4_4b i.sexPreg i.oral_v_dhp
stset DiedFU15, fail(DiedNext14) id(obsno)
/*split time*/
stsplit new, at(4 8)

loc varlist _ISpeciesX_2    ///
            _ISpeciesX_4    ///
            _ISpeciesX_5    ///
            _IEthnicX_2     ///
            _IEthnicX_3     ///
            _IAGR4_4b_1     ///
            _IAGR4_4b_2     ///
            _IAGR4_4b_3     ///
            _IsexPreg_2     ///
            _IsexPreg_3     ///
            _Ioral_v_dh_1


foreach i of varlist `varlist' {
gen tv3`i' = `i' * (new==4)
gen tv7`i' = `i' * (new==8)
}



/*List of variables for model including interactions with time */
loc varlist _ISpeciesX_2                                        ///
            _ISpeciesX_4                                        ///
            _ISpeciesX_5    tv3_ISpeciesX_5     tv7_ISpeciesX_5 ///
            _IEthnicX_2                                         ///
            _IEthnicX_3                                         ///
            _IAGR4_4b_1                                         ///
            _IAGR4_4b_2     tv3_IAGR4_4b_2       tv7_IAGR4_4b_2 ///
            _IAGR4_4b_3                                         ///
            _IsexPreg_2                                         ///
            _IsexPreg_3                                         ///
            _Ioral_v_dh_1

/*run model*/
stcox `varlist' if ip==0 , efron  allbaselevels vsquish cluster(hrn) cformat(%6.2f) nolog

/*save and store estimates and scaled Schoenfeld residuals for later access*/
estimates
estimates store M3bTV37_24oct2015
estimates save  M3bTV37_24oct2015
predict sch_M3bTV37*, sca
save MalEps_v1.9.3_M3b_TV37.dta
linktest, cluster(hrn) efron
estat phtest, d

*Model 4
/*load data*/
use "C:\data\malaria\MalEps_v1.9.3_r5oct2015.dta", clear
/*create indicator variables*/
xi i.SpeciesX i.EthnicX i.AGR4_4b i.sexPreg i.ivArt
/*declare survival time*/
stset DiedFU15, fail(DiedNext14) id(obsno)
/*split time*/
stsplit new, at(8)
/*list of variables to create potential splits for*/
loc varlist _ISpeciesX_2    ///
            _ISpeciesX_4    ///
            _ISpeciesX_5    ///
            _IEthnicX_2     ///
            _IEthnicX_3     ///
            _IAGR4_4b_1     ///
            _IAGR4_4b_2     ///
            _IAGR4_4b_3     ///
            _IsexPreg_2     ///
            _IsexPreg_3     ///
            _IivArt_1


foreach i of varlist `varlist' {
gen tv7`i' = `i' * (new==8)
}


/*list of variables including TVCs*/
loc varlist _ISpeciesX_2                        ///
            _ISpeciesX_4                        ///
            _ISpeciesX_5                        ///
            _IEthnicX_2                         ///
            _IEthnicX_3                         ///
            _IAGR4_4b_1     tv7_IAGR4_4b_1      ///
```

```
            _IAGR4_4b_2                         ///
            _IAGR4_4b_3     tv7_IAGR4_4b_3      ///
            _IsexPreg_2                         ///
            _IsexPreg_3                         ///
            _IivArt_1

*run Cox model  with TVCs
stcox `varlist' if ip==1 , efron allbaselevels vsquish cluster(hrn)  cformat(%6.2f) nolog
*store results for later access
estimates
estimates store M4aTV37_24oct2015
estimates save  M4aTV37_24oct2015

predict sch_M4aTV37*, sca

save MalEps_v1.9.3_M4a_TV37.dta

linktest, cluster(hrn) efron
estat phtest, d




****Coefficient plots
*Model 2
cd "C:\data\malaria\results\efron 24 Oct 2015 10_47_59\"
use "C:\data\malaria\results\efron 24 Oct 2015 10_47_59\MalEps_v1.9.3_M2a_TV37.dta", clear
estimates use M2aTV37_24oct2015.ster
/*reload estimation sample for stored model*/
estimates esample:  ///
   _ISpeciesX_2     ///
tv3_ISpeciesX_2     ///
tv7_ISpeciesX_2     ///
   _ISpeciesX_4     ///
   _ISpeciesX_5     ///
    _IEthnicX_2     ///
    _IEthnicX_3     ///
 tv3_IEthnicX_3     ///
 tv7_IEthnicX_3     ///
    _IAGR4_4b_1     ///
    _IAGR4_4b_2     ///
 tv3_IAGR4_4b_2     ///
 tv7_IAGR4_4b_2     ///
    _IAGR4_4b_3     ///
 tv7_IAGR4_4b_3     ///
    _IsexPreg_2     ///
 tv7_IsexPreg_2     ///
    _IsexPreg_3     ///
 tv3_IsexPreg_3     ///
 tv7_IsexPreg_3     ///
  _Ioral_v_dh_1

  estimates


  label variable  _ISpeciesX_2  "{it:P.vivax}"
  label variable  _ISpeciesX_4  "{it:P.malariae}"
  label variable  _ISpeciesX_5  "mixed"
  label variable  _IEthnicX_2   "Lowland"
  label variable  _IEthnicX_3   "non-Papuan"
  label variable  _IAGR4_4b_1   "0 to {&lt} 1"
  label variable  _IAGR4_4b_2   "1 to {&lt} 5 "
  label variable  _IAGR4_4b_3   "5 to {&lt} 15"
  label variable  _IsexPreg_2   "female (pregnant)"
  label variable  _IsexPreg_3   "male"
  label variable  _Ioral_v_dh_1 "DHP"

  label variable  tv3_ISpeciesX_2  "{it:t}{sub:3}{&rarr}{it:t}{sub:7} "
  label variable  tv3_ISpeciesX_4  "{it:t}{sub:3}{&rarr}{it:t}{sub:7} "
  label variable  tv3_ISpeciesX_5  "{it:t}{sub:3}{&rarr}{it:t}{sub:7} "
  label variable  tv3_IEthnicX_2   "{it:t}{sub:3}{&rarr}{it:t}{sub:7} "
  label variable  tv3_IEthnicX_3   "{it:t}{sub:3}{&rarr}{it:t}{sub:7} "
  label variable  tv3_IAGR4_4b_1   "{it:t}{sub:3}{&rarr}{it:t}{sub:7} "
  label variable  tv3_IAGR4_4b_2   "{it:t}{sub:3}{&rarr}{it:t}{sub:7} "
  label variable  tv3_IAGR4_4b_3   "{it:t}{sub:3}{&rarr}{it:t}{sub:7} "
  label variable  tv3_IsexPreg_2   "{it:t}{sub:3}{&rarr}{it:t}{sub:7} "
  label variable  tv3_IsexPreg_3   "{it:t}{sub:3}{&rarr}{it:t}{sub:7} "
  label variable  tv3_Ioral_v_dh_1 "{it:t}{sub:3}{&rarr}{it:t}{sub:7} "

  label variable  tv7_ISpeciesX_2  "{it:t}{sub:7}{&rarr}{it:t}{sub:14}"
  label variable  tv7_ISpeciesX_4  "{it:t}{sub:7}{&rarr}{it:t}{sub:14}"
  label variable  tv7_ISpeciesX_5  "{it:t}{sub:7}{&rarr}{it:t}{sub:14}"
  label variable  tv7_IEthnicX_2   "{it:t}{sub:7}{&rarr}{it:t}{sub:14}"
  label variable  tv7_IEthnicX_3   "{it:t}{sub:7}{&rarr}{it:t}{sub:14}"
  label variable  tv7_IAGR4_4b_1   "{it:t}{sub:7}{&rarr}{it:t}{sub:14}"
  label variable  tv7_IAGR4_4b_2   "{it:t}{sub:7}{&rarr}{it:t}{sub:14}"
```

```
  label variable  tv7_IAGR4_4b_3   "{it:t}{sub:7}{&rarr}{it:t}{sub:14}"
  label variable  tv7_IsexPreg_2   "{it:t}{sub:7}{&rarr}{it:t}{sub:14}"
  label variable  tv7_IsexPreg_3   "{it:t}{sub:7}{&rarr}{it:t}{sub:14}"
  label variable  tv7_Ioral_v_dh_1 "{it:t}{sub:7}{&rarr}{it:t}{sub:14}"


coefplot    (M2a_24oct2015,     mc("241 163 64")  ciopts(lc("241 163 64"))                          ///
                label(multivariable model 2) )                                                    ///
            (M2aTV37_24oct2015, mc("153 142 195") ciopts(lc("153 142 195"))                        ///
                label("model 2 with time interaction"))                                           ///
            , eform baselevels xline(1, lc("27 158 119"))                                         ///
        order(    _ISpeciesX_2 tv3_ISpeciesX_2 tv7_ISpeciesX_2 _ISpeciesX_4 _ISpeciesX_5         ///
                . _IEthnicX_2 _IEthnicX_3 tv3_IEthnicX_3 tv7_IEthnicX_3                           ///
                . _IAGR4_4b_1 _IAGR4_4b_2 tv3_IAGR4_4b_2 tv7_IAGR4_4b_2 _IAGR4_4b_3 tv7_IAGR4_4b_3  ///
                . _IsexPreg_2 tv7_IsexPreg_2 _IsexPreg_3 tv3_IsexPreg_3 tv7_IsexPreg_3            ///
                . _Ioral_v_dh_1)                                                                  ///
        headings(_ISpeciesX_2 = "{it:P.falciparum} (reference)"                                   ///
            _IEthnicX_2 = "Highland (reference)"                                                  ///
            _IAGR4_4b_1 = " {&ge} 15 (reference)"                                                 ///
            _IsexPreg_2 = "female, pregnant (reference)"                                          ///
            _Ioral_v_dh_1 = "oral quinine (reference)")                                           ///
            coeflabels(,labsize(small)) legend(cols(1))                                           ///
            graphr(color(white) lc(white) margin(2 2 0 0)) plotr(color(white) lc(white))          ///
            grid(within glwidth(thin)) ysize(20) xsize(15)                                        ///
        xtitle("Hazard Ratio", margin(medsmall))    xlab(,labsize(small))                         ///
        xmlab(1 "reference",add tlcolor("27 158 119") labcolor("27 158 119"))                     ///
        subtitle("Early admission in outpatients on oral treatment",                              ///
        size(medium) margin(-30 0 2 0))                                                           ///
    note("Note: {it:t} refers to analysis time in days since presentation with a malaria episode; " ///
    "           i.e. time is split at day 3 and/or day 7 following presentation, where specified."  ///
    , margin(-37 0 0 2) size(vsmall) )

    graph export "C:\data\malaria\figures\Model2_compare24oct2015.emf", as(emf) replace

**Graph piece-wise regression of model 2 incorporating split at days 3 and 7 following the day of presentation
estimates
matrix M2tv = r(table)'
di "Day 0 to Day 3: HR"  %9.2f M2tv[1,1] %9.2f M2tv[1,5] %9.2f M2tv[1,6]
di "Day 3 to Day 7: HR"  %9.2f M2tv[2,1] %9.2f M2tv[2,5] %9.2f M2tv[2,6]
di "Day 7 to Day 14: HR" %9.2f M2tv[3,1] %9.2f M2tv[3,5] %9.2f M2tv[3,6]
local hr1 = M2tv[1,1]
local hr2 = M2tv[2,1]
local hr3 = M2tv[3,1]
local hr1_lci = M2tv[1,5]
local hr1_uci = M2tv[1,6]
local hr2_lci = M2tv[2,5]
local hr2_uci = M2tv[2,6]
local hr3_lci = M2tv[3,5]
local hr3_uci = M2tv[3,6]
di "Day 0 to Day 3: " %9.2f `hr1' %9.2f `hr1_lci' %9.2f `hr1_uci'
di "Day 3 to Day 7: " %9.2f `hr2' %9.2f `hr2_lci' %9.2f `hr2_uci'
di "Day 7 to Day 14:" %9.2f `hr3' %9.2f `hr3_lci' %9.2f `hr3_uci'
twoway      function y = `hr1', range(2  4) lwidth(thick) lpattern(solid) lc("217 95 2")    ||  ///
            function y = `hr2', range(4  8) lwidth(thick) lpattern(solid) lc("217 95 2")    ||  ///
            function y = `hr3', range(8 15) lwidth(thick) lpattern(solid) lc("217 95 2")    ||  ///
            function y = `hr1_lci', range(2  4) lpattern(dash) lc("253 205 172")            ||  ///
            function y = `hr2_lci', range(4  8) lpattern(dash) lc("253 205 172")            ||  ///
            function y = `hr3_lci', range(8 15) lpattern(dash) lc("253 205 172")            ||  ///
            function y = `hr1_uci', range(2  4) lpattern(dash) lc("253 205 172")            ||  ///
            function y = `hr2_uci', range(4  8) lpattern(dash) lc("253 205 172")            ||  ///
            function y = `hr3_uci', range(8 15) lpattern(dash) lc("253 205 172")            ||  ///
            function y = 1, lwidth(thick) lpattern(solid) range(2 15) lc("27 158 119")      ||  ///
            function y = 0.92, lpattern(solid) range(2 15) lc("247 247 247") lwidth(thick)      ///
            legend(order(10 "{it:P.falciparum} (reference)"                                     ///
                        11 "{it:P.vivax} multivariable model 2, HR 0.92"                        ///
                         1 "{it:P.vivax} model 2 with time interaction, HR (95% CI)")           ///
                    pos(6) col(1))                                                              ///
            xtitle("Time (days) since presentation with malaria", margin(medsmall))             ///
            ytitle("Hazard Ratio", margin(medsmall))                                            ///
            ylab(`hr1' `hr2' `hr3' 0.65 1 2, nogrid angle(h) labsize(small) format(%9.2f))      ///
            xlab(2 "1" 4 "3" 8 "7"  15 "14", labsize(small)) xmtick(1(1)15)                     ///
            xscale(nofextend)    yscale(log fextend)                                            ///
            graphr(color(white) lc(white)) plotr(color(white) lc(white))


*Model 3
cd "C:\data\malaria\results\efron 24 Oct 2015 10_47_59\"
use "C:\data\malaria\results\efron 24 Oct 2015 10_47_59\MalEps_v1.9.3_M3b_TV37.dta", clear
estimates use M3bTV37_24oct2015.ster

estimates esample: ///
   _ISpeciesX_2 ///
   _ISpeciesX_4 ///
   _ISpeciesX_5 ///
   tv3_ISpeciesX_5 ///
   tv7_ISpeciesX_5 ///
```

```
    _IEthnicX_2 ///
    _IEthnicX_3 ///
    _IAGR4_4b_1 ///
    _IAGR4_4b_2 ///
 tv3_IAGR4_4b_2 ///
 tv7_IAGR4_4b_2 ///
    _IAGR4_4b_3 ///
    _IsexPreg_2 ///
    _IsexPreg_3 ///
  _Ioral_v_dh_1

  estimates


  label variable  _ISpeciesX_2  "{it:P.vivax}"
  label variable  _ISpeciesX_4  "{it:P.malariae}"
  label variable  _ISpeciesX_5  "mixed"
  label variable  _IEthnicX_2   "Lowland"
  label variable  _IEthnicX_3   "non-Papuan"
  label variable  _IAGR4_4b_1   "0 to {&lt} 1"
  label variable  _IAGR4_4b_2   "1 to {&lt} 5 "
  label variable  _IAGR4_4b_3   "5 to {&lt} 15"
  label variable  _IsexPreg_2   "female (pregnant)"
  label variable  _IsexPreg_3   "male"
  label variable  _Ioral_v_dh_1 "DHP"


  label variable  tv3_ISpeciesX_2  "{it:t}{sub:3}{&rarr}{it:t}{sub:7} "
  label variable  tv3_ISpeciesX_4  "{it:t}{sub:3}{&rarr}{it:t}{sub:7} "
  label variable  tv3_ISpeciesX_5  "{it:t}{sub:3}{&rarr}{it:t}{sub:7} "
  label variable  tv3_IEthnicX_2   "{it:t}{sub:3}{&rarr}{it:t}{sub:7} "
  label variable  tv3_IEthnicX_3   "{it:t}{sub:3}{&rarr}{it:t}{sub:7} "
  label variable  tv3_IAGR4_4b_1   "{it:t}{sub:3}{&rarr}{it:t}{sub:7} "
  label variable  tv3_IAGR4_4b_2   "{it:t}{sub:3}{&rarr}{it:t}{sub:7} "
  label variable  tv3_IAGR4_4b_3   "{it:t}{sub:3}{&rarr}{it:t}{sub:7} "
  label variable  tv3_IsexPreg_2   "{it:t}{sub:3}{&rarr}{it:t}{sub:7} "
  label variable  tv3_IsexPreg_3   "{it:t}{sub:3}{&rarr}{it:t}{sub:7} "
  label variable  tv3_Ioral_v_dh_1 "{it:t}{sub:3}{&rarr}{it:t}{sub:7} "

  label variable  tv7_ISpeciesX_2  "{it:t}{sub:7}{&rarr}{it:t}{sub:14}"
  label variable  tv7_ISpeciesX_4  "{it:t}{sub:7}{&rarr}{it:t}{sub:14}"
  label variable  tv7_ISpeciesX_5  "{it:t}{sub:7}{&rarr}{it:t}{sub:14}"
  label variable  tv7_IEthnicX_2   "{it:t}{sub:7}{&rarr}{it:t}{sub:14}"
  label variable  tv7_IEthnicX_3   "{it:t}{sub:7}{&rarr}{it:t}{sub:14}"
  label variable  tv7_IAGR4_4b_1   "{it:t}{sub:7}{&rarr}{it:t}{sub:14}"
  label variable  tv7_IAGR4_4b_2   "{it:t}{sub:7}{&rarr}{it:t}{sub:14}"
  label variable  tv7_IAGR4_4b_3   "{it:t}{sub:7}{&rarr}{it:t}{sub:14}"
  label variable  tv7_IsexPreg_2   "{it:t}{sub:7}{&rarr}{it:t}{sub:14}"
  label variable  tv7_IsexPreg_3   "{it:t}{sub:7}{&rarr}{it:t}{sub:14}"
  label variable  tv7_Ioral_v_dh_1 "{it:t}{sub:7}{&rarr}{it:t}{sub:14}"


coefplot    (M3b_24oct2015,     mc("241 163 64")  ciopts(lc("241 163 64"))                          ///
                label(multivariable model 3) )                                                    ///
            (M3bTV37_24oct2015, mc("153 142 195") ciopts(lc("153 142 195"))                       ///
                label("model 3 with time interaction"))                                           ///
            , eform baselevels xline(1, lc("27 158 119"))                                         ///
        order(    _ISpeciesX_2 _ISpeciesX_4 _ISpeciesX_5 tv3_ISpeciesX_5 tv7_ISpeciesX_5         ///
                . _IEthnicX_2 _IEthnicX_3                                                         ///
                . _IAGR4_4b_1 _IAGR4_4b_2 tv3_IAGR4_4b_2 tv7_IAGR4_4b_2 _IAGR4_4b_3              ///
                . _IsexPreg_2 _IsexPreg_3                                                         ///
                . _Ioral_v_dh_1)                                                                  ///
        headings(_ISpeciesX_2 = "{it:P.falciparum} (reference)"                                   ///
            _IEthnicX_2 = "Highland (reference)"                                                  ///
            _IAGR4_4b_1 = " {&ge} 15 (reference)"                                                 ///
            _IsexPreg_2 = "female, pregnant (reference)"                                          ///
            _Ioral_v_dh_1 = "oral quinine (reference)")                                           ///
            coeflabels(,labsize(small)) legend(cols(1))                                           ///
            graphr(color(white) lc(white) margin(2 2 0 0)) plotr(color(white) lc(white))          ///
            grid(within glwidth(thin)) ysize(20) xsize(15)                                        ///
        xtitle("Hazard Ratio", margin(medsmall))    xlab(,labsize(small))                         ///
        xmlab(1 "reference",add tlcolor("27 158 119") tlength(*8) labcolor("27 158 119") tlwidth(medium))   ///
        subtitle("Early death in outpatients on oral treatment",                                  ///
        size(medium) margin(-30 0 2 0))                                                           ///
    note("Note: {it:t} refers to analysis time in days since presentation with a malaria episode; " ///
    "          i.e. time is split at day 3 and/or day 7 following presentation, where specified."  ///
    , margin(-37 0 0 2) size(vsmall) )

    graph export "C:\data\malaria\figures\Model3_compare24oct2015.emf", as(emf) replace

*Model 4 comparison of with and without time split
cd "C:\data\malaria\results\efron 24 Oct 2015 10_47_59\"
use "C:\data\malaria\results\efron 24 Oct 2015 10_47_59\MalEps_v1.9.3_M4a_TV37.dta", clear
estimates use M4aTV37_24oct2015.ster
/*reload estimation sample for stored model*/
estimates esample: ///
```

```
   _ISpeciesX_2 ///
   _ISpeciesX_4 ///
   _ISpeciesX_5 ///
    _IEthnicX_2 ///
    _IEthnicX_3 ///
    _IAGR4_4b_1 ///
 tv7_IAGR4_4b_1 ///
    _IAGR4_4b_2 ///
    _IAGR4_4b_3 ///
 tv7_IAGR4_4b_3 ///
    _IsexPreg_2 ///
    _IsexPreg_3 ///
  _IivArt_1

  estimates

  label variable  _ISpeciesX_2  "{it:P.vivax}"
  label variable  _ISpeciesX_4  "{it:P.malariae}"
  label variable  _ISpeciesX_5  "mixed"
  label variable  _IEthnicX_2   "Lowland"
  label variable  _IEthnicX_3   "non-Papuan"
  label variable  _IAGR4_4b_1   "0 to {&lt} 1"
  label variable  _IAGR4_4b_2   "1 to {&lt} 5 "
  label variable  _IAGR4_4b_3   "5 to {&lt} 15"
  label variable  _IsexPreg_2   "female (pregnant)"
  label variable  _IsexPreg_3   "male"
  * label variable _Ioral_v_dh_1 "DHP"
  label variable  _IivArt_1     "artesunate"

  label variable  tv7_ISpeciesX_2  "{it:t}{sub:7}{&rarr}{it:t}{sub:14}"
  label variable  tv7_ISpeciesX_4  "{it:t}{sub:7}{&rarr}{it:t}{sub:14}"
  label variable  tv7_ISpeciesX_5  "{it:t}{sub:7}{&rarr}{it:t}{sub:14}"
  label variable  tv7_IEthnicX_2   "{it:t}{sub:7}{&rarr}{it:t}{sub:14}"
  label variable  tv7_IEthnicX_3   "{it:t}{sub:7}{&rarr}{it:t}{sub:14}"
  label variable  tv7_IAGR4_4b_1   "{it:t}{sub:7}{&rarr}{it:t}{sub:14}"
  label variable  tv7_IAGR4_4b_2   "{it:t}{sub:7}{&rarr}{it:t}{sub:14}"
  label variable  tv7_IAGR4_4b_3   "{it:t}{sub:7}{&rarr}{it:t}{sub:14}"
  label variable  tv7_IsexPreg_2   "{it:t}{sub:7}{&rarr}{it:t}{sub:14}"
  label variable  tv7_IsexPreg_3   "{it:t}{sub:7}{&rarr}{it:t}{sub:14}"
  * label variable tv7_Ioral_v_dh_1 "{it:t}{sub:7}{&rarr}{it:t}{sub:14}"
  label variable  tv7_IivArt_1        "{it:t}{sub:7}{&rarr}{it:t}{sub:14}"

***Graph
coefplot    (M4a_24oct2015,     mc("241 163 64")  ciopts(lc("241 163 64"))                ///
                label(multivariable model 4) )                                          ///
            (M4aTV37_24oct2015, mc("153 142 195") ciopts(lc("153 142 195"))               ///
                label("model 4 with time interaction"))                                 ///
            , eform baselevels xline(1, lc("27 158 119"))                               ///
       order(    _ISpeciesX_2 _ISpeciesX_4 _ISpeciesX_5                                 ///
               . _IEthnicX_2 _IEthnicX_3                                                ///
               . _IAGR4_4b_1   tv7_IAGR4_4b_1   _IAGR4_4b_2  _IAGR4_4b_3 tv7_IAGR4_4b_3 ///
               . _IsexPreg_2 _IsexPreg_3                                                ///
               . _IivArt_1)                                                             ///
       headings(_ISpeciesX_2 = "{it:P.falciparum} (reference)"                          ///
           _IEthnicX_2 = "Highland (reference)"                                         ///
           _IAGR4_4b_1 = " {&ge} 15 (reference)"                                        ///
           _IsexPreg_2 = "female, pregnant (reference)"                                 ///
           _IivArt_1 = "IV quinine (reference)")                                        ///
           coeflabels(,labsize(small)) legend(cols(1))                                  ///
           graphr(color(white) lc(white) margin(2 2 0 0)) plotr(color(white) lc(white)) ///
           grid(within glwidth(thin)) ysize(20) xsize(15)                               ///
       xtitle("Hazard Ratio", margin(medsmall))    xlab(,labsize(small))                ///
       xmlab(1 "reference",add tlcolor("27 158 119")                                    ///
       tlength(*8) labcolor("27 158 119") tlwidth(medium))                              ///
       subtitle("Early death in outpatients on intravenous treatment",                  ///
       size(medium) margin(-30 0 2 0))                                                  ///
   note("Note: {it:t} refers to analysis time in days since presentation with a malaria episode; " ///
   "          i.e. time is split at day 3 and/or day 7 following presentation, where specified."  ///
   , margin(-37 0 0 2) size(vsmall) )

   graph export "C:\data\malaria\figures\Model3_compare24oct2015.emf", as(emf) replace
```