```
/*******************************************************************************
* Malaria early morbidity and mortality                                        *
* Main Models - October 2015                                                   *
* - first run do files '1a...' (Setup) and '1b...' (stvary diagnostics)        *
*******************************************************************************/


*Set up log and working directory
capture log close
version 13.1
set linesize 100
set more off
cd "C:\Users\Carl\Google Drive\MPH\Projects\Malaria project\Data\results"
loc today = c(current_date)
log using "malariaproject_log_`today'.txt", append text

*local macro to establish method of Cox model ties handling
loc ties efron

* prepare folder for results
local T = c(current_time)
local T = subinstr("`T'",":","_",.)
mkdir "`ties' `today' `T'"
cd "`ties' `today' `T'"

*timer Start

timer clear 1
timer on 1


*** Admission
*Logistic regression

    * ****Logistic Univariable
    set more off
    loc AgeCond if AgeGr7d ==1
    loc varlist i.SpeciesX i.EthnicX i.AGR4_4b i.sexPreg  i.whiteCat
    foreach v of loc varlist {
    logistic ip `v' `AgeCond', allbaselevels vsquish cluster(hrn)  cformat(%6.2f) nolog
    }

    *m1a
    logistic ip i.SpeciesX i.Ethnic ib4.AGR4 i.sexPreg, vce(cluster hrn)
    estimates store M1a_OR_v2   /* store model for later retrieval */
    estimates save M1a_OR_v2
    linktest
    estat gof

    *m1b
    logistic ip i.SpeciesX i.Ethnic ib4.AGR4 i.sexPreg i.whiteCat, vce(cluster hrn)
    estimates store M1b_OR  /* store model for later retrieval */
    estimates save M1b_OR
    linktest

* Wald test of inclusion of WBC count normality (since likelihood ratio test is inappropriate with clustered data

test  1.whiteCat  2.whiteCat


*Cox PH regression
    stset AdmFU15, fail(AdmNext14) id(obsno)

    *Cox PH univariable (Model 2, admission)
    loc varlist i.SpeciesX i.AGR4_4b i.EthnicX i.sexPreg i.oral_v_dhp i.whiteCat
    foreach v of loc varlist {
    stcox `v' if ip==0, allbaselevels vsquish cluster(hrn) efron cformat(%6.2f) nolog
    }


    *m2a: risk of admission within 15 days in those who were not admitted immediately, and on oral or dhp
    stcox i.SpeciesX ib4.AGR4 i.Ethnic i.sexPreg i.oral_v_dhp if ip==0, cluster(hrn) efron
    estimates store M2a_HR  /* store model for later retrieval */
    estimates save M2a_HR
    linktest, cluster(hrn) efron

    *m2b: as above with wcc, limited to those with laboratory data
    stcox i.SpeciesX ib4.AGR4 i.Ethnic i.sexPreg i.oral_v_dhp i.whiteCat if ip==0, cluster(hrn) efron
    estimates store M2b_HR      /* store model for later retrieval */
    estimates save M2b_HR
    linktest, cluster(hrn) efron

    * Wald test of inclusion of WBC count normality

    test  1.whiteCat  2.whiteCat
```

```
    *Mann-Whitney U-test
    ranksum MalPres, by(oral_v_dhp)
    ranksum MalPres, by(oral_v_dhp) porder
    median MalPres, by(oral_v_dhp) exact

***Death
    stset DiedFU15, fail(DiedNext14) id(obsno)


    *m3a:  risk of death by day 15 in those who were not initially admitted, including oral / dhp first treatment
    stcox i.SpeciesX ib4.AGR4 i.Ethnic i.sexPreg i.oral_v_dhp if ip==0, cluster(hrn) efron
    estimates store M3a_HR          /* store model for later retrieval */
    estimates save  M3a_HR
    linktest, cluster(hrn) efron

    *m3b: with WBC count normality
    stcox i.SpeciesX ib4.AGR4 i.Ethnic i.sexPreg i.oral_v_dhp i.whiteCat if ip==0, cluster(hrn) efron
    estimates store M3b_HR          /* store model for later retrieval */
    estimates save  M3b_HR
    linktest, cluster(hrn) efron

    * Wald test of inclusion of WBC count normality

    test  1.whiteCat  2.whiteCat

    * m4a: risk of death by day 15 limited to those who were admitted immediately & rx'd IV treatment first


    stcox i.SpeciesX ib4.AGR4_4b i.Ethnic i.sexPreg i.ivArt if ip==1, cluster(hrn) efron
    estimates store M4a_HR      /* store model for later retrieval */
    estimates save M4a_HR
    linktest, cluster(hrn) efron

    *m4b:  with WBC count normality
    stcox i.SpeciesX ib4.AGR4 i.Ethnic i.sexPreg i.ivArt i.whiteCat if ip==1, cluster(hrn) efron
    estimates store M4b_HR          /* store model for later retrieval */
    estimates save M4b_HR
    linktest, cluster(hrn) efron

    * Wald test of inclusion of WBC count normality

    test  1.whiteCat  2.whiteCat

    timer off 1
    timer list 1
```